
Abstract

This paper presents a COQ formalization of linear algebra over elementary

divisor rings, that is, rings where every matrix is equivalent to a matrix in Smith

normal form. The main results are the formalization that these rings support

essential operations of linear algebra, the classification theorem of finitely pre-

sented modules over such rings and the uniqueness of the Smith normal form

up to multiplication by units. We present formally verified algorithms comput-

ing this normal form on a variety of coefficient structures including Euclidean

domains and constructive principal ideal domains. We also study different ways

to extend Bézout domains in order to be able to compute the Smith normal form

of matrices. The extensions we consider are: adequacy (i.e. the existence of

a gdco operation), Krull dimension ≤ 1 and well-founded strict divisibility.

1 Introduction

The goal of this paper is to develop linear algebra for elementary divisor rings, that

is, rings where there is an algorithm for computing the Smith normal form of ma-

trices. The algorithms we show to compute this normal form can be seen as a

generalization of Gaussian elimination that can, in particular, be defined for the

ring of integers. The main source of inspiration for this work is the formalization

of finite dimensional vector spaces by Georges Gonthier [16] in which spaces are

represented using matrices and all subspace constructions can be elegantly defined

from Gaussian elimination. This enables a concrete and point-free presentation

of linear algebra which is suitable for formalization as it takes advantage of the

small scale reflection methodology of the SSREFLECT extension and the Mathemat-

ical Components library [17] for the COQ proof assistant [8]. When generalizing

this to elementary divisor rings there are two essential problems that need to be

resolved before the theory may be formalized:

1. What is a suitable generalization of finite dimensional vector spaces when

considering more general classes of rings than fields as coefficients?

2. What rings are elementary divisor rings?

A possible answer to the first problem is finitely generated R-modules, i.e. finite di-

mensional vector spaces with coefficients in a general ring instead of a field. How-

ever these are not as well behaved as finite dimensional vector spaces as there might

be relations among the generators. In other words, not all finitely generated mod-

ules are free. To overcome this, we restrict our attention further and consider finitely

presented modules, which are modules specified by a finite number of generators

and a finite number of relations between these. This class of modules may be repre-

sented concretely using matrices, which in turn means that we can apply the same

approach as in [16] and implement all operations by manipulating the presentation

matrices.

A standard answer to the second problem is principal ideal domains like the

ring of integers (denoted by Z) and the ring of univariate polynomials over a field

(denoted by k[x]). The classical definition of principal ideal domains is integral

domains where all ideals are principal (i.e. generated by one element). In partic-

ular it means that principal ideal domains are Noetherian as all ideals are finitely

generated. Classically this is equivalent to the ascending chain condition for ideals,

however in order to prove this equivalence classical reasoning is used in essential

ways. In fact, if these definitions are read constructively they are so strong that no

ring except the trivial ring satisfies them [32]. Principal ideal domains are hence

problematic from a constructive point of view as they are Noetherian.

A possible solution is to restrict the attention to Euclidean domains (which in-

clude both Z and k[x]) and show how to compute the Smith normal form of matri-

ces over these rings. This approach is appealing as it allows for a simple definition

of the Smith normal form algorithm that resembles the one of Gaussian elimination.

While Euclidean domains are important, we would like to be more general. In order

to achieve this we consider an alternative approach that is customary in constructive

algebra: to generalize all statements and not assume Noetheriannity at all [25]. If

we do this for principal ideal domains we get Bézout domains, which are rings where

every finitely generated ideal is principal. However, it is an open problem whether

all Bézout domains are elementary divisor rings or not [26]. Hence we study differ-

ent assumptions that we can add to Bézout domains in order to prove that they are

elementary divisor rings. The properties we define and study independently are:

1. Adequacy (i.e. the existence of a gdco operation);

2. Krull dimension ≤ 1;

3. Strict divisibility is well-founded.

The last one can be seen as a constructive approximation to the ascending chain

condition for principal ideals, so this kind of Bézout domains will be referred to as

constructive principal ideal domains.

The main contributions of this paper are the formalization1, using the COQ proof

assistant with the SSREFLECT extension, of:

• Rings with explicit divisibility, GCD domains, Bézout domains, constructive

principal ideal domains and Euclidean domains (section 2);

• An algorithm computing the Smith normal form of matrices with coefficients

in Euclidean domains and the generalization to constructive principal ideal

domains (section 3);

• Linear algebra over elementary divisor rings and the classification theorem

for finitely presented modules over elementary divisor rings (section 4);

• Proofs that Bézout domains extended with one of the three extensions above

are elementary divisor rings and how these notions are related (section 5);

The paper ends with an overview of related work (section 6), followed by conclu-

sions and future work (section 7).

2 Rings with explicit divisibility

In this section we recall definitions and basic properties of rings with explicit di-

visibility, GCD domains, Bézout domains, constructive principal ideal domains and

Euclidean domains.

2.1 Rings with explicit divisibility

Throughout the paper all rings are discrete integral domains, i.e. commutative

rings with a unit, decidable equality and no zero divisors. This section is loosely

based on the presentation of divisibility in discrete domains of Mines, Richman and

Ruitenberg in [29]. The central notion we consider is:

Definition 2.1 A ring R has explicit divisibility if it has a divisibility test that pro-

duces witnesses.

That is, given a and b we can test if a | b and if this is the case get x such that

b = xa. Two elements a, b ∈ R are associates if a | b and b | a, which is equivalent

to b = ua for some unit u because we have cancellation. Note that this gives rise

to an equivalence relation. This notion will play an important role later as we will

1The formal development can be found at: https://github.com/CoqEAL/CoqEAL

show that the Smith normal form of a matrix is unique up to multiplication by units,

that is, up to associated elements.

A GCD domain is an example of a ring with explicit divisibility:

Definition 2.2 A GCD domain R is a ring with explicit divisibility in which every pair

of elements has a greatest common divisor, that is, for a, b ∈ R there is gcd(a, b) such

that gcd(a, b) | a, gcd(a, b) | b and ∀g, (g | a)∧ (g | b)→ g | gcd(a, b).

Note first that we make no restriction on a and b, so they can both be zero.

In this case the greatest common divisor is zero. This makes sense as zero is the

maximum element for the divisibility relation. Note also that as R is assumed to be

a ring with explicit divisibility we get that gcd(a, b) | a means that there is a′ such

that a = a′ gcd(a, b). By Euclid’s algorithm we know that both Z and k[x] are GCD

domains.

With the above definition the greatest common divisor of two elements is not

necessarily unique, e.g. the greatest common divisor of 2 and 3 inZ is either 1 or−1.

But if we consider equality up to multiplication by units (i.e. up to associatedness)

the greatest common divisor is unique, so in the rest of the paper equality will

denote equality up to associatedness when talking about the gcd.

Most of the rings we will study in this paper are Bézout domains:

Definition 2.3 A Bézout domain is a GCD domain R such that for any two elements

a, b ∈ R there is x , y ∈ R such that ax + b y = gcd(a, b).

Let a and b be two elements in a ring R. If R is a GCD domain we can compute

g = gcd(a, b) together with witnesses to the ideal inclusion (a, b) ⊆ (g). Further,

if R is a Bézout domain we can compute witnesses for the inclusion (g) ⊆ (a, b) as

well. This can be generalized to multiple elements a1, . . . , an ∈ R to obtain witnesses

for the inclusions (a1, . . . , an) ⊆ (g) and (g) ⊆ (a1, . . . , an) where g is the greatest

common divisor of the ai . Bézout domains can hence be characterized as rings in

which every finitely generated ideal is principal, which means that they are non-

Noetherian generalizations of principal ideal domains.

Note that, on the one hand there exists a′ and b′ such that a = a′g and b = b′g,

and on the other hand we have x and y such that ax + b y = g. Therefore, by

dividing with g, we obtain a Bézout relation between a′ and b′, namely a′x+a′ y =

1.

This definition can be extended to give a constructive version of principal ideal

domains. We say that a divides b strictly if a | b but b ∤ a, then we can define:

Definition 2.4 A constructive principal ideal domain is a Bézout domain in which

the strict divisibility relation is well-founded.

By well-founded we mean that any descending chain of strict divisions is finite.

This can be seen as a constructive approximation to the ascending chain condition

for principal ideals and hence to Noetheriannity. Both Z and k[x] can be proved to

be Bézout domains and satisfy the condition of constructive principal ideal domains.

In fact, this can be done for any ring on which the extended Euclidean algorithm

can be implemented. These rings are called Euclidean domains:

Definition 2.5 A Euclidean domain is a ring R with a Euclidean norm N : R→ N

such that for any a ∈ R and nonzero b ∈ R we have N (a)¶N (ab). Further, for any

a ∈ R and nonzero b ∈ R we can find q, r ∈ R such that a = bq + r and either r = 0

or N (r)<N (b).

In the case of Z and k[x] we can take respectively the absolute value function

and the degree function as Euclidean norm. Then the standard division algorithms

for these rings can be used to compute q and r.

2.2 Formalization of algebraic structures

The algebraic structures have been formalized in the same manner as in the SS-

REFLECT library [15] using packed classes (implemented by mixins and canonical

structures). We will now discuss the formalization of these new structures starting

with the definition of rings with explicit divisibility:

Inductive div_spec (R : ringType) (a b :R) : option R -> Type :=

| DivDvd x of a = x * b : div_spec a b (Some x)

| DivNDvd of (forall x, a != x * b) : div_spec a b None.

Record mixin_of R := Mixin {

div : R -> R -> option R;

_ : forall a b, div_spec a b (div a b)

}.

This structure is denoted by DvdRing and for a ring to be an instance it needs

to have a function div that returns an option type, such that if div a b = None

then a ∤ b, and if div a b = Some x then x is the witness that a | b. The notation

used for div a b in the formalization is a %/? b. There is also a %| notation for the

div function that returns a boolean, this relies on a coercion from option to bool

defined in the SSREFLECT libraries (mapping None to false and Some x to true for

any x). Using this we have implemented the notion of associatedness, denoted by

%=, and the basic theory of divisibility.

Next we have the GCDDomain structure which is implemented as:

Record mixin_of R := Mixin {

gcd : R -> R -> R;

_ : forall d a b, (d %| gcd a b) = (d %| a) && (d %| b)

}.

For a ring to be a GCDDomain it needs to have a gcd function satisfying the prop-

erty above. This property is sufficient as it implicitly gives that gcd(a, b) | a and

gcd(a, b) | b since divisibility is reflexive.

The BezoutDomain structure looks like:

Inductive bezout_spec (R : gcdDomainType) (a b : R) : R * R -> Type :=

BezoutSpec x y of gcdr a b %= x * a + y * b : bezout_spec a b (x, y).

Record mixin_of R := Mixin {

bezout : R -> R -> (R * R);

_ : forall a b, bezout_spec a b (bezout a b)

}.

Recall that a constructive principal ideal domain is a Bézout domain where strict

divisibility is well-founded. This is denoted by PID and is implemented by:

Definition sdvdr (R : dvdRingType) (x y : R) := (x %| y) && ~~(y %| x).

Record mixin_of R := Mixin {

_ : well_founded (@sdvdr R)

}.

The notation x %<| y will be used to denote sdvdr x y. We will see more precisely

in section 3.2 how well_founded is defined formally in COQ’s standard library when

we use it to prove the termination of our Smith normal form algorithm.

We also have the EuclideanDomain structure that represents Euclidean domains:

Inductive edivr_spec (R : ringType)

(g : R -> nat) (a b : R) : R * R -> Type :=

EdivrSpec q r of a = q * b + r & (b != 0) ==> (g r < g b)

: edivr_spec g a b (q, r).

Record mixin_of R := Mixin {

enorm : R -> nat;

ediv : R -> R -> R * R;

_ : forall a b, a != 0 -> enorm b <= enorm (a * b);

_ : forall a b, edivr_spec enorm a b (ediv a b)

}.

This structure contains the Euclidean norm and the Euclidean division func-

tion together with their proofs of correctness. We have implemented the extended

version of Euclid’s algorithm for Euclidean domains and proved that it satisfies

bezout_spec. Hence we get that Euclidean domains are Bézout domains. We have

also proved that any EuclideanDomain is a PID which means that strict divisibility

is well-founded in both Z and k[x].

The relationship between the algebraic structures presented in this section can

be depicted by:

EuclideanDomain ⊂ PID ⊂ BezoutDomain ⊂ GCDDomain ⊂ DvdRing ⊂ IntegralDomain

where IntegralDomain is already present in the SSREFLECT hierarchy. In the next

section we consider an algorithm for computing the Smith normal form of matrices

over the first two algebraic structures in the chain of inclusions. This means that

these two structures are elementary divisor rings. In section 5 we will generalize to

Bézout domains of Krull dimension ≤ 1 and adequate domains that fit in between

PID and BezoutDomain in the chain of inclusions.

3 A verified algorithm for the Smith Normal Form

In [24] Kaplansky introduced the notion of elementary divisor rings as rings where

every matrix is equivalent to a matrix in Smith normal form, that is, given a m× n

matrix M there exist invertible matrices P and Q of size m×m and n×n respectively,

such that PMQ = D where D is a diagonal matrix of the form:





















d1 0 · · · · · · 0

...
...

0 dk 0 · · · 0
... 0 0

...
...

...
. . .

...

0 · · · 0 · · · · · · 0





















with the additional property that di | di+1 for all i.

Let us first explain how we formalized the notion of Smith normal form in COQ,

with the following representation of matrices taken from the SSREFLECT library:

Inductive matrix R m n := Matrix of {ffun 'I_m * 'I_n -> R}.

Here 'I_m is the type of ordinals (i.e. natural numbers bounded by m) which has

exactly m inhabitants and can be coerced to nat. Matrices are then implemented as

finite functions over finite sets of indices, with dependent types being used to ensure

well-formedness. We use the notation 'M[R]_(m,n) for the type matrix R m n, the

notation 'rV[R]_m for the type of row vectors of length m and the notation 'cV[R]_m

for column vectors of height m. The ring R is often omitted from these notations

when it can be inferred from the context.

In order to express that a matrix is in Smith normal form, we define diag_mx_seq,

which rebuilds a diagonal matrix from a list (note that the type of lists is called seq

in the SSREFLECT library) of diagonal coefficients:

Definition diag_mx_seq m n (s : seq R) :=

\matrix_(i < m, j < n) s`_i *+ (i == j :> nat).

The notation x *+ n, where x belongs to a ring and n is a natural number, stands

for the sum x + . . .+ x iterated n times. In the expression of the general coeffi-

cients of the matrix above, i and j are ordinals of type 'I_m and 'I_n respectively.

The notation i == j :> nat tells COQ to compare them as natural numbers and

returns a boolean. A coercion then sends this boolean to a natural number (true

is interpreted by 1 and false by 0). Thus s`_i *+ (i == j :> nat) denotes the

element of index i in s if i and j have the same value, 0 otherwise.

Now if M is a matrix, an algorithm for computing the Smith normal form should

return a list s and two matrices P and Q such that:

• The sequence s is sorted for the divisibility relation.

• The matrix diag_mx_seq m n s is equivalent to M, with transition matrices P

and Q.

Which translates formally to an inductive predicate:

Inductive smith_spec R m n M : 'M[R]_m * seq R * 'M[R]_n -> Type :=

SmithSpec P d Q of P *m M *m Q = diag_mx_seq m n d

& sorted %| d

& P \in unitmx

& Q \in unitmx : smith_spec M (P,d,Q).

We have packaged this in the same manner as above in order to represent ele-

mentary divisor rings:

Record mixin_of R := Mixin {

smith : forall m n, 'M[R]_(m,n) -> 'M[R]_m * seq R * 'M[R]_n;

_ : forall m n (M : 'M[R]_(m,n)), smith_spec M (smith M)

}.

In the rest of this section we will see direct proofs that Euclidean domains and

constructive principal ideal domains provide instances of this structure.

3.1 Smith normal form over Euclidean domains

We mentioned in the introduction that constructive finite dimensional linear algebra

over a field can be reduced to matrix encodings. Information like the rank and

determinant is then reconstructed from the encoding using Gaussian elimination,

which involves three kinds of operations on the matrix:

1. Swapping two rows (resp. columns)

2. Multiplying one row (resp. column) by a nonzero constant

3. Adding to a row (resp. column) the product of another one by a constant

These three operations are interesting because they are compatible with matrix

equivalence. In particular, they can be expressed as left (resp. right) multiplication

by invertible matrices.

The same algorithm fails to apply in general to a matrix over a ring, since it may

require a division by the pivot, which could be not exact. The content of this section

can thus be seen as a generalization of Gaussian elimination to Euclidean domains.

To make this extension possible, a new kind of elementary operations needs to

be introduced. Let a and b be elements of a Euclidean domain R. Bézout’s identity

gives u and v such that ua + vb = γ where γ = gcd(a, b). Let us note a′ = a
γ and

b′ = b
γ , these divisions being exact by definition of the gcd. We get the identity:

ua′ + vb′ = 1. Consider the following square matrix of size n:

EBezout(a, b, n, k) =



























(col. k)

u v

1

...

1

(row k) −b′ a′

1

...

1



























The coefficients not explicitly shown in EBezout are assumed to be zeros. Note

that det(EBezout(a, b, n, k)) = ua′ + vb′ = 1, so in particular the matrix above is

invertible.

We formalize these matrices as follows:

Definition combine_mx (a b c d : R) (m : nat) (k : 'I_m) :=

let k' := lift 0 k in

let d := \row_j (a *+ (j == 0) + d *+ (j == k') +

((j != 0) && (j != k'))%:R) in

diag_mx d + c *: delta_mx k' 0 + b *: delta_mx 0 k'.

Definition Bezout_mx (a b : R) (m : nat) (k : 'I_m) :=

let:(_,u,v,a1,b1) := egcdr a b in combine_mx u v (-b1) a1 k.

For an ordinal i of type 'I_m, lift 0 i represents the ordinal 1 + i of type 'I_(1 + m).

The notation \row_(j < m) r j corresponds to the row matrix [r 0, ..., r (m-1)],

if the dimension can be automatically inferred then we can just write \row_j r j.

If b is a boolean, the term b%:R reduces to 1 if b is true, 0 otherwise. The matrix

diag_mx d correspond to the diagonal matrix where diagonal coefficients are the

coefficients of the row matrix d, and delta_mx i j is the matrix which has only ze-

ros except at position (i, j), where the coefficient is 1. Finally, a *: A is the matrix

A multiplied by the scalar a. Note that the Bézout identity between a and b is given

by the function egcdr, which is exported by the underlying Euclidean ring.

Like other elementary operations, the left product by EBezout(a, b, n, k) can be

interpreted as an operation on the rows:

EBezout(a, b, n, k)×



























L1

L2

...

Lk−1

Lk

Lk+1

...

Ln



























=



























uL1 + vLk

L2

...

Lk−1

−b′L1 + a′Lk

Lk+1

...

Ln

These row operations are described formally by:

Definition combine_step (a b c d : R) (m n : nat)

(M : 'M_(1 + m,1 + n)) (k : 'I_m) :=

let k' := lift 0 k in

let r0 := a *: row 0 M + b *: row k' M in

let rk := c *: row 0 M + d *: row k' M in

\matrix_i (r0 *+ (i == 0) + rk *+ (i == k') +

row i M *+ ((i != 0) && (i != k'))).

Definition Bezout_step (a b : R) (m n : nat)

(M : 'M_(1 + m, 1 + n)) (k : 'I_m) :=

let:(_,u,v,a1,b1) := egcdr a b in combine_step u v (-b1) a1 M k.

Here row i M represents the i:th row of M. A lemma connects these row operations

to the corresponding elementary matrices:

Lemma Bezout_stepE a b (m n : nat) (M : 'M_(1 + m,1 + n)) k :

Bezout_step a b M k = Bezout_mx a b k *m M.

Let now M = (ai, j) be a matrix with coefficients in R. We will now show how to

reduce M to its Smith normal form using elementary operations. As for Gaussian

elimination, we start by finding a nonzero pivot g in M , which is moved to the

upper-left corner (if M = 0, M is in Smith normal form). We search the first column

for an element which is not divisible by g. Let us assume that g ∤ ak,1, we then

multiply the matrix on the left by EBezout(g, ak,1, n, k) :

EBezout(g, ak,1, n, k)×



















g L1

a2,1 L2

...
...

ak,1 Lk

...
...

an,1 Ln



















=



















γ uL1 + vLk

a2,1 L2

...

−g ′g + a′ak,1 −g ′L1 + a′Lk

...
...

an,1 Ln



















with the Bézout identity ug + vak,1 = γ = gcd(g, ak,1) and posing as previously

g ′ =
g

γ , we have a′ =
ak,1

γ .

By definition of γ, we have: γ | −g ′g + a′ak,1. Moreover, all the coefficients in

the first column of M which were divisible by g are also by γ. We can therefore

repeat this process until we get a matrix whose upper-left coefficient (which we

still name g) divides all the coefficients in the first column. Linear combinations on

rows can thence lead to a matrix B of the following shape:

B =

g b1,2 · · · b1,n

g
...

...

...
...

...

g bm,2 · · · bm,n





































We then search the whole framed submatrix of B for an element that is not

divisible by g. If such a coefficient bi, j is found, it is moved to the top by permuting

rows 1 and i. Thus g is still the upper-left coefficient2 and multiplications on the

right by EBezout matrices allow, like previously, to obtain a matrix whose upper-left

coefficient divides all the others.

This first step is implemented by the function improve_pivot_rec:

1 Fixpoint improve_pivot_rec k {m n} :

2 'M[R]_(1 + m) -> 'M[R]_(1 + m, 1 + n) -> 'M[R]_(1 + n) ->

3 'M[R]_(1 + m) * 'M[R]_(1 + m, 1 + n) * 'M[R]_(1 + n) :=

4 match k with

5 | 0 => fun P M Q => (P,M,Q)

6 | p.+1 => fun P M Q =>

7 let a := M 0 0 in

8 if find1 M a is Some i then

9 let Mi0 := M (lift 0 i) 0 in

10 let P := Bezout_step a Mi0 P i in

11 let M := Bezout_step a Mi0 M i in

12 improve_pivot_rec p P M Q

13 else

14 let u := dlsubmx M in let vM := ursubmx M in let vP := usubmx P in

15 let u' := map_mx (fun x => 1 - odflt 0 (x %/? a)) u in

16 let P := col_mx (usubmx P) (u' *m vP + dsubmx P) in

17 let M := block_mx a%:M vM

18 (const_mx a) (u' *m vM + drsubmx M) in

19 if find2 M a is Some (i,j) then

20 let M := xrow 0 i M in let P := xrow 0 i P in

21 let a := M 0 0 in

22 let M0ij := M 0 (lift 0 j) in

23 let Q := (Bezout_step a M0ij Q^T j)^T in

24 let M := (Bezout_step a M0ij M^T j)^T in

25 improve_pivot_rec p P M Q

26 else (P, M, Q)

27 end.

2This trick has been inspired to the authors by a proof-oriented formalization of a similar algorithm

by Georges Gonthier.

If A, B, C and D are four matrices (with matching dimensions) then block_mx A B C D

is the matrix:

M =

�

A B

C D

�

where A = ulsubmx M, B = ursubmx M, C = dlsubmx M and D = drsubmx M. Sim-

ilarly C = col_mx A B is a column matrix with A = usubmx C and B = dsubmx C

(the functions for constructing and destructing row matrices have similar names).

The matrix const_mx a is the matrix where each coefficient is equal to a and xrow i j M

is the matrix M with the rows i and j exchanged.

The function improve_pivot_rec takes as arguments a natural number k which

represents the number of remaining steps, the original matrix and two current tran-

sition matrices. If the number of remaining steps is zero, the matrices are returned

unchanged (line 5). If not, the first column is searched for an element that is not

divisible by the pivot (function find1, line 8). If such an element is found on a row

of index i, a Bézout step is performed between the first row and the one of index

i, and the function is called recursively (lines 9 to 12). If, on the contrary, the pivot

divides all the elements in the first column, some linear combinations (lines 14

to 18) bring us back to a matrix of the shape of the matrix B seen above. Finally,

the remaining lines search the whole matrix for an element that is not divisible by

the pivot (function find2), perform a Bézout step on the columns if appropriate,

and call the function recursively.

We have made several choices when implementing this function. First, the ar-

gument k bounding the number of steps makes it easy to have a structural recursion

(this natural number decreases by 1 at each step). In this usual technique, k is often

called the fuel of the recursion. The flip side is that in order to call the function, an

a priori bound on the number of steps has to be provided. It is at this point that the

hypothesis we made that R is a Euclidean domain comes in handy: we can take as

a bound the Euclidean norm of the upper-left coefficient of the original matrix.

We also chose to abstract over initial transition matrices, which are updated

as the process goes on. From a computational standpoint, this approach has two

benefits. First, it avoids the need for products by transition matrices, asymptotically

more costly than to perform the elementary operations directly. Then, it makes

the function improve_pivot_rec tail-recursive, which can have a good impact on

performance.

The flip side is that it is slightly more difficult to express and manipulate for-

mally the link between the matrices taken as arguments and those returned by the

function. Indeed, the specification of this function involves inverses of transition

matrices:

Inductive improve_pivot_rec_spec m n P M Q :

'M_(1 + m) * 'M_(1 + m,1 + n) * 'M_(1 + n) -> Type :=

ImprovePivotRecSpec P' M' Q' of

P^-1 *m M *m Q^-1 = P'^-1 *m M' *m Q'^-1

& (forall i j, M' 0 0 %| M' i j)

& (forall i, M' i 0 = M' 0 0)

& M' 0 0 %| M 0 0

& P' \in unitmx

& Q' \in unitmx : improve_pivot_rec_spec P M Q (P',M',Q').

The statement above can be read as follows: given three matrices P, M and Q, a

triple (P,M',Q') satisfies the specification if applying to M the inverse of elementary

operations represented by the initial transition matrices P and Q gives the same

result as applying the inverses of the transition matrices P' and Q' to M'.

The correctness lemma of the function improve_pivot_rec states that for an

initial matrix M whose upper-left coefficient is nonzero and has a norm smaller than

a natural number k, and for invertible matrices P and Q, the triple returned by

improve_pivot_rec k P M Q satisfies the specification represented by the induc-

tive type improve_pivot_rec_spec:

Lemma improve_pivot_recP k m n (P : 'M_(1 + m)) (M : 'M_(1 + m,1 + n)) Q :

enorm (M 0 0) <= k -> M 0 0 != 0 ->

P \in unitmx -> Q \in unitmx ->

improve_pivot_rec_spec P M Q (improve_pivot_rec k P M Q).

Initially, we call the function improve_pivot_rec with identity transition matrices:

Definition improve_pivot k m n (M : 'M_(1 + m, 1 + n)) :=

improve_pivot_rec k 1%:M M 1%:M.

By successive subtractions of the first row from all the others and then by linear

combinations of columns, we get a matrix C:

C =









g 0 · · · 0

0
... C ′

0









where g divides all coefficients of C ′.

The global algorithm computing the Smith normal form proceeds as follows: it

stores the pivot g obtained after the previous step, then divides all coeficients of C ′

by g and is applied recursively to the resulting matrix. Let us pose k = min(m, n).

From the pivots g1, . . . , gk obtained, the final output of the algorithm is given by the

following sequence d1, . . . , dk:

d1, d2, . . . , dk = g1, g1 g2, . . . ,

k
∏

i=1

gi

The Smith normal form of the original matrix is then the following diagonal matrix

of size m× n:






















d1

d2

. . .

dk

0

...

0























This global procedure is implemented by the function Smith :

1 Fixpoint Smith {m n} : 'M[R]_(m,n) -> 'M[R]_(m) * seq R * 'M[R]_(n) :=

2 match m, n return 'M[R]_(m, n) -> 'M[R]_(m) * seq R * 'M[R]_(n) with

3 | _.+1, _.+1 => fun M : 'M[R]_(1 + _, 1 + _) =>

4 if find_pivot M is Some (i, j) then

5 let a := M i j in let M := xrow i 0 (xcol j 0 M) in

6 let: (P,M,Q) := improve_pivot (enorm a) M in

7 let a := M 0 0 in

8 let u := dlsubmx M in let v := ursubmx M in

9 let v' := map_mx (fun x => odflt 0 (x %/? a)) v in

10 let M := drsubmx M - const_mx 1 *m v in

11 let: (P', d, Q') := Smith (map_mx (fun x => odflt 0 (x %/? a)) M) in

12 (lift0_mx P' *m block_mx 1 0 (- const_mx 1) 1 *m (xcol i 0 P),

13 a :: [seq x * a | x <- d],

14 (xrow j 0 Q) *m block_mx 1 (- v') 0 1 *m lift0_mx Q')

15 else (1%:M, [::], 1%:M)

16 | _, _ => fun M => (1%:M, [::], 1%:M)

17 end.

If M has type 'M[R]_n then lift0_mx M = block_mx 1 0 0 M of type 'M[R]_(1 + n).

The notation [seq f x | x <- xs] is like a list comprehension in HASKELL and

means map f xs.

The function Smith takes as argument a matrix and returns a sequence made

of the nonzero diagonal coefficients of its Smith form, as well as the corresponding

transition matrices. The first step (lines 4 and 5) consists in searching for a nonzero

pivot in the whole matrix and moving it in the upper-left position. If no pivot is

found, all the coefficients are zero and an empty sequence is therefore returned.

Otherwise, the function improve_pivot defined previously is called (line 6), then

some elementary row operations are performed (lines 8 to 10) to get a matrix of the

shape of the matrix C shown above. The bottom-right submatrix is then divided by

the pivot and a recursive call is performed (line 11). The sequence of coefficients

and transition matrices obtained are then updated (lines 12 to 14).

We have stated and proved the following correctness lemma:

Lemma SmithP (m n : nat) (M : 'M_(m,n)) : smith_spec M (Smith M).

Using this we have instantiated the structure of elementary divisor rings on Eu-

clidean domains.

3.2 Extension to principal ideal domains

We mentioned in section 2 that (constructive) principal ideal domains were Bézout

domains with a well-founded divisibility relation. Well-foundedness is defined in

COQ’s standard library using an accessibility predicate [30]:

Inductive Acc (A : Type) (R : A -> A -> Prop) (x : A) : Prop :=

Acc_intro : (forall y : A, R y x -> Acc R y) -> Acc R x.

The idea is that all objects of the inductive type Acc have to be built by a finite

number of applications of the constructor Acc_intro. Hence, for any a such that

Acc R a, all chains (xn) such that R xn+1 xn and x0 = a have to be finite. Note

however that there can be infinitely many elements x such that R x a. Using this

definition of accessibility, we can now state that a relation over a type A is well-

founded if all elements in A are accessible:

Definition well_founded (A : Type) (R : A -> A -> Prop) :=

forall a, Acc R a.

Remember that in the previous section, we used the hypothesis that the ring of

coefficients was Euclidean when we computed an a priori bound on the number of

steps the function improve_pivot needed to perform. To extend the algorithm to

principal ideal domains, we replace the recursion on this bound with a well-founded

induction on the divisibility relation.

Fixpoint improve_pivot_rec m n (P : 'M_(1 + m)) (M : 'M_(1 + m, 1 + n))

(Q : 'M_(1 + n)) (k : Acc (@sdvdr R) (M 0 0)) :

'M_(1 + m) * 'M_(1 + m, 1 + n) * 'M_(1 + n) :=

match k with Acc_intro IHa =>

if find1P M (M 0 0) is Pick i Hi then

let Ai0 := M (lift 0 i) 0 in

let P := Bezout_step (M 0 0) Ai0 P i in

improve_pivot_rec P Q (IHa _ (sdvd_Bezout_step Hi))

else

let u := dlsubmx M in let vM := ursubmx M in let vP := usubmx P in

let u' := map_mx (fun x => 1 - odflt 0 (x %/? M 0 0)) u in

let P := col_mx (usubmx P) (u' *m vP + dsubmx P) in

let A := block_mx (M 0 0)%:M vM

(const_mx (M 0 0)) (u' *m vM + drsubmx M) in

if find2P A (M 0 0) is Pick (i,j) Hij then

let A := xrow 0 i A in

let P := xrow 0 i P in

let a := A 0 0 in

let A0j := A 0 (lift 0 j) in

let Q := (Bezout_step a A0j Q^T j)^T in

improve_pivot_rec P Q (IHa _ (sdvd_Bezout_step2 Hij))

else (P, A, Q)

end.

The main difference with the function improve_pivot defined in section 3.1 is

that we need to prove that the upper-left element of the matrix on which we make

the recursive call is strictly smaller than the one of the original matrix. To build

these proofs, we use the functions find1P and find2P which have more expressive

(dependent) types than their counterparts find1 and find2 that we used previously.

They return not only an element of the matrix given as argument, but also a proof

that the pivot does not divide this element.

This proof is then used to show that the upper-left coefficient of the matrix

decreases, thanks to the following two lemmas:

Lemma sdvd_Bezout_step m n (M : 'M_(1 + m,1 + n)) (k : 'I_m) :

~~ (M 0 0 %| M (lift 0 k) 0) ->

(Bezout_step (M 0 0) (M (lift 0 k) 0) M k) 0 0 %<| M 0 0.

Lemma sdvd_Bezout_step2 m n i j u' vM (M : 'M[R]_(1 + m, 1 + n)) :

let B : 'M_(1 + m, 1 + n) :=

block_mx (M 0 0)%:M vM (const_mx (M 0 0)) (u' *m vM + drsubmx M) in

let C := xrow 0 i B in

~~ (M 0 0 %| B i (lift 0 j)) ->

(Bezout_step (C 0 0) (C 0 (lift 0 j)) C^T j)^T 0 0 %<| M 0 0.

Now, to define the improve_pivot function, we use the hypothesis sdvdr_wf

that the divisibility relation is well-founded:

Definition improve_pivot m n (M : 'M_(1 + m, 1 + n)) :=

improve_pivot_rec 1 1 (sdvdr_wf (M 0 0)).

The function Smith of section 3.1 is essentially unchanged, the only difference

being that we removed the first argument of improve_pivot (which was an a priori

bound on the number of steps of improve_pivot_rec).

We have shown how to compute the Smith normal form on Euclidean domains

and more generally on principal ideal domains. In the next section, we will explain

how to develop a constructive theory of linear algebra based on the existence of

such an algorithm.

4 Elementary divisor rings

The goal of this section is to develop some theory about linear algebra over ele-

mentary divisor rings and discuss the formalization of the classification theorem for

finitely presented modules over these rings.

4.1 Linear algebra over elementary divisor rings

One of the key operations in linear algebra is to compute solutions to systems of

equations. A suitable algebraic setting for doing so is rings where every finitely

generated ideal is finitely presented. These rings are called coherent:

Definition 4.1 A ring is coherent if for any matrix M it is possible to compute a

matrix L such that:

X M = 0 ↔ ∃Y. X = Y L

This means that L generates the module of solutions of X M = 0, i.e. that L

generates the kernel of M . The notion of coherent rings is usually not mentioned in

classical presentations of algebra since Noetherian rings are automatically coherent,

but in a computationally meaningless way. It is however a fundamental notion, both

conceptually [25, 29] and computationally [1, 2]. Coherent rings have previously

been represented in COQ [9] so we will not discuss the details of the formalization

here. Instead we show that elementary divisor rings are coherent.

Let M be a m× n matrix with coefficients in an elementary divisor rings. There

are invertible matrices P and Q such that PMQ = D where D is a diagonal matrix

in Smith normal form. The rank of M , denoted r(M), is the number of nonzero

elements of D. The kernel of M can be computed by:

ker(M) = (Im − Ir(M))P

where Im is a m×m identity matrix and Ir(m) is a m×m partial identity matrix with

r(M) ones on the diagonal and then zeros. The idea behind this definition is that:



















0

... 0

0

1

0
...

1





































d1

. . . 0

dk

0

0
...

0



















= 0

So ker(M)MQ = 0 and since Q is invertible, we have ker(M)M = 0. We can

implement the rank operator and state its correctness by:

Definition mxrank m n (M : 'M[R]_(m,n)) :=

let: (P,d,Q) := smith M in size [seq x <- d | x != 0 ].

Definition kermx m n (M : 'M[R]_(m,n)) : 'M[R]_m :=

let: (P,d,Q) := smith M in copid_mx (mxrank M) *m P.

Lemma kermxP m n (M : 'M[R]_(m,n)) (X : 'rV[R]_m) :

reflect (exists Y : 'rV[R]_m, X = Y *m kermx M) (X *m M == 0).

where copid_mx corresponds to the partial identity matrix. The reflect statement

should be read as: the boolean equality X *m M == 0 holds if and only if there exists

Y : 'rV[R]_m such that X = Y *m kermx M.

An algorithm computing the cokernel of a matrix can be implemented in a sim-

ilar fashion. This way we have implemented a small library inspired by the one

on matrix algebra for fields of SSREFLECT [16], but based on Smith normal form

instead of Gaussian elimination.

Another important notion in constructive algebra is strongly discrete rings:

Definition 4.2 A ring is strongly discrete if membership in finitely generated ideals

is decidable and if whenever x ∈ (x1, . . . , xn), there exists y1, . . . , yn such that x =
∑

i x i yi .

If a ring is both coherent and strongly discrete it is not only possible to solve

homogeneous systems of equations but also arbitrary systems of the kind X M = B

where X is a m× n matrix, M a n× k matrix and B a nonzero m× k matrix.

It is easy to see that Bézout domains are strongly discrete as any finitely gener-

ated ideal is principal. To test if x ∈ (a1, . . . , an) first compute a principal ideal (g)

equivalent to (a1, . . . , an) and then test if g | x . If this is the case we may construct

the witness and otherwise we know that x /∈ (a1, . . . , an).

It is also straightforward to prove that any elementary divisor ring is a Bézout

domain. Given a, b ∈ R we can compute the Smith normal form of a row matrix

containing a and b. This gives us an invertible 1× 1 matrix P, an invertible 2× 2

matrix Q, and g ∈ R such that:

P
�

a b
�

Q =
�

g 0
�

As P and Q are invertible we get that g is the greatest common divisor of a and

b. The Bézout coefficients are then found by performing the matrix multiplications

on the left-hand side of the equality. Hence we get that elementary divisor rings are

not only coherent but also strongly discrete.

In section 5 we consider extensions to Bézout domains that make them elemen-

tary divisor rings and hence form a good setting for doing linear algebra. The next

subsection shows that the existence of an algorithm for computing the Smith nor-

mal form makes finitely presented modules over elementary divisor rings especially

well-behaved.

4.2 Finitely presented modules over elementary divisor rings

Recall that a module is said to be finitely presented if it can be described using a

finite set of generators and a finite set of relations among these. A convenient way

to express this is:

Definition 4.3 An R-moduleM is finitely presented if there is an exact sequence:

Rm1 Rm0 M 0
M π

This means that π is a surjection and M a matrix representing the m1 relations

among the m0 generators of the moduleM . Another way to think ofM is as the

cokernel of M , that is, M ≃ coker(M) = Rm0/Im(M). So a module has a finite

presentation if it can be expressed as the cokernel of a matrix. As all information

of finitely presented modules is contained in its presentation matrix we get that

all algorithms on finitely presented modules can be described by manipulating the

presentation matrices [11, 18, 25].

A morphism ϕ between finitely presented modulesM and N given by presen-

tations:

Rm1 Rm0 M 0 Rn1 Rn0 N 0
M N

is represented by a m0×n0 matrixϕG and a m1×n1 matrixϕR such that the following

diagram commutes:

Rm1 Rm0 M 0

Rn1 Rn0 N 0

M

ϕR ϕG ϕ

N

The intuition why two matrices are needed is that the morphism affects both

the generators and relations of the modules, hence the names ϕG and ϕR. In this

paper we adopt the SSREFLECT convention that composition is read in diagrammatic

order (i.e. from left to right) when writing equations obtained from commutative

diagrams. This means that the equation related to the above diagram is written

MϕG = ϕRN .

In order for us to be able to compute kernels of morphisms we need to assume

that the underlying ring is coherent so that we can solve systems of equations in-

volving the underlying matrices. If the underlying ring is also strongly discrete, it is

possible to represent morphisms using only ϕG and a proof that ∃X .X N = MϕG as

any system of equations of the kind X M = B is solvable. Two of the authors have

previously [7] formalized finitely presented modules over coherent and strongly

discrete rings in COQ which provides a basis for this part of the formalization.

It is in general not possible to decide if two finitely presented modules are iso-

morphic or not. However, if the underlying ring is an elementary divisor ring, it

becomes possible. Indeed, let R be an elementary divisor ring and M be a m1 ×m0

matrix presenting an R-moduleM . As M is equivalent to a diagonal matrix D, there

are invertible matrices P and Q such that MQ = P−1D. This gives a commutative

diagram:

Rm1 Rm0 M 0

Rm1 Rm0 D 0

M

P−1 Q ϕ

D

We can further prove that ϕ is an isomorphism as P and Q are invertible, and

hence get thatM ≃D ≃ coker(D). Now, since D is a diagonal matrix with nonzero

elements d1, . . . , dn ∈ R on the diagonal, we get that:

M ≃ Rm0−n ⊕ R/(d1)⊕ · · · ⊕ R/(dn) (1)

with the additional property that di | di+1 for all 1 ¶ i < n. Note that if di is a

unit then R/(di)≃ 0. This means that the theory of finitely presented modules over

elementary divisor rings R is particularly well-behaved as any finitely presented

R-module M can be decomposed into a direct sum of a free module and cyclic

modules. This is the first part of the classification theorem for finitely presented

modules over elementary divisor rings, the second part is the fact that the di are

unique up to multiplication by units which makes the decomposition unique.

The uniqueness part is also necessary in order to get a decision procedure for

the isomorphism of finitely presented modules over elementary divisor rings. So far

we only know that any module may be decomposed as above, but there is, a priori,

no reason why two isomorphic modules should have related decompositions.

In the next section we will see that the Smith normal form is unique up to

multiplication by units if the underlying ring has a gcd operation, which in turn

completes the classification theorem and gives us a decision procedure for module

isomorphism.

4.3 Uniqueness of the Smith normal form

The formal proof that the Smith normal form is unique up to multiplication by units

presented here is based on [4]. In order to formalize this proof we need to represent

minors (determinants of submatrices) in COQ. This notion was defined in a previ-

ous work on formalizing the Sasaki-Murao algorithm computing the characteristic

polynomial of a matrix [10]. With the SSREFLECT definition of matrices it is easy

to give a definition of submatrices (denoted by M( f , g)) and minors:

Definition submatrix m n p q (f : 'I_p -> 'I_m) (g : 'I_q -> 'I_n)

(M : 'M[R]_(m,n)) : 'M[R]_(p,q) :=

\matrix_(i < p, j < q) M (f i) (g j).

Definition minor m n p (f : 'I_p -> 'I_m) (g : 'I_p -> 'I_n)

(M : 'M[R]_(m,n)) : R := \det (submatrix f g M).

For example, the rows (resp. columns) of the matrix M( f , g) are the rows (resp.

columns) f (0), f (1), ... (resp. g(0), g(1), ...) of M . It would be natural to define

submatrices only when f and g are strictly increasing, however this is not necessary

as many theorems are true for arbitrary functions. We denote p in the definition of

minor above as the order of the minor, that is, a minor of order p is the determinant

of a submatrix of dimension p× p.

The key result in order to prove the uniqueness theorem for the Smith normal

form is that the product of the k first elements of the diagonal in the Smith normal

form is associated to the gcd of the minors of order k of the original matrix. More

precisely, let M be the original matrix and di the i:th element of the diagonal in

the Smith normal form of M , also let ~mk be the minors of order k of M , then the

statement is:

k
∏

i=1

di = gcd( ~mk)

Using the big operators library of SSREFLECT [3] this can be expressed compactly

as:

Lemma Smith_gcdr_spec :

\prod_(i < k) d`_i %= \big[gcdr/0]_f \big[gcdr/0]_g minor f g M.

The order of the minors that we consider are given by the type of f and g. For

the sake of readability, we have omitted these types.

The first step in proving this is by showing that it holds for the Smith normal

form of M , namely the diagonal matrix D. Since it is a diagonal matrix, the only

nonzero minors of order k are the determinants of diagonal matrices of dimension

k×k, that are products of k elements of the diagonal of D. Also, since each element

of the diagonal divides the next one, the greatest common divisor of the minors of

order k is the product of the k first elements of the diagonal. For example, if the

diagonal is (a, b, c) with a | b and b | c then gcd(ab, bc, ac) = ab.

The next step is to prove that the gcd of the minors of order k of M are associated

to the gcd of the minors of D (which we already know is associated to the product

of the elements on the diagonal). To prove this it suffices to show that these two

divide each other, as the proofs in both directions are very similar we only show

that the gcd of the minors of order k of M divides the gcd of the minors of order k

of D.

By definition, x divides gcd(~y) if and only if x divides every y in ~y . So we must

show that the gcd of the minors of order k of M divides each minor of order k of the

diagonal matrix D. Now, there are invertible matrices P and Q such that PMQ = D.

Hence we must show that gcd( ~mk) divides det((PMQ)( f , g)) for all f and g. The

right-hand side is the determinant of a product of matrices of different sizes whose

product is square, which can be simplified with the Binet-Cauchy formula:

det(MN) =
∑

I∈P ({1,...,l})
#|I |=k

det(MI )det(NI )

where M is a k× l matrix and N is a l × k matrix. MI (resp. NI ) is the matrix of the

k columns (resp. rows) with indices in I .

The formalization of this formula builds on the work in [10] and follows Zeng’s

proof presented in [36]. Note that the standard determinant identity for products

of square matrices of the same size follows as a special case of the above formula.

Once again the theorem can be expressed compactly using the big operators of SS-

REFLECT:

Lemma BinetCauchy :

\det (M *m N) = \sum_(f : {ffun 'I_k -> 'I_l} | strictf f)

((minor id f M) * (minor f id N)).

Here the sum is taken over all strictly increasing functions from {1, . . . , k} to

{1, . . . , l}. We require the functions to be strictly increasing so that the minors that

we consider in the sum correspond to the mathematical concept of minor.

This theorem makes it possible for us to transform det((PMQ)( f , g)) to a sum

of minors and, once again, it suffices to show that gcd( ~mk) divides each of the

summands. Hence, after some simplifications, we must show that for all h and i

we have:

\big[gcdr/0]_f \big[gcdr/0]_g minor f g M %| minor h i M

which is true by definition of the gcd. Note that it is not necessary to require that f

and g are strictly increasing. Indeed, if they are not, there are two cases:

• Either f or g is not injective and so minor f g M = 0.

• If both f and g are injective there exist permutations r and s such that f' = f \o r

and g' = g \o s are strictly increasing. As the permutation of rows or columns

of a matrix just leads to the determinant being multiplied by the signature of

the permutation we get minor f g M %= minor f' g' M.

But for all a we have gcd(a, 0) = a and gcd(a, a) = a, so in each case the terms

corresponding to the minors obtained from not strictly increasing f and g does not

change the value of the gcd of the minors.

Now if the above result is applied with k = 1, the uniqueness of the first di-

agonal element is proved, and then by induction all of the diagonal elements are

showed to be unique (up to multiplication by units). This means that for any matrix

M equivalent to a diagonal matrix D in Smith normal form, each of the diagonal

elements of the Smith normal form of M will be associate to the corresponding diag-

onal element in D. The uniqueness of the Smith normal form is expressed formally

as follows:

Lemma Smith_unicity m n (M : 'M[R]_(m,n)) (d : seq R) :

sorted %| d -> equivalent M (diag_mx_seq m n d) ->

forall i, i < minn m n -> (smith_seq M)`_i %= d`_i.

Hence we have proved that the Smith normal form is unique up to multiplica-

tion by units. This gives a test to know if two matrices are equivalent. Indeed, since

the Smith normal form of a matrix is equivalent to it, two matrices are equivalent if

and only if they have the same normal form. Moreover, we know that the decompo-

sition in equation (1) is unique up to multiplication by units. Hence we get a way

for deciding if two finitely presented modules are isomorphic or not: compute the

Smith normal form of the presentation matrices and then test if they are equivalent

up to multiplication by units.

This concludes the classification theorem for finitely presented modules over

elementary divisor rings. It can be seen as a constructive version of the classifica-

tion theorem for finitely generated modules over principal ideal domains. Classical

proofs of this use the fact that a principal ideal domain R is Noetherian which im-

plies that any R-module is coherent, i.e. that any finitely generated module is also

finitely presented. But this proof has no computational content (see exercise 3 in

chapter III.2 of [29]), so instead we have to restrict to finitely presented modules.

In section 3.2 we showed that (constructive) principal ideal domains are elemen-

tary divisor rings which gives us the classical result in the case of finitely presented

modules. In the next section we will prove that more general classes of rings than

principal ideal domains are elementary divisor rings which gives more instances of

the classification theorem.

5 Extensions to Bézout domains that are elementary

divisor rings

As mentioned in the introduction, it is an open problem whether all Bézout domains

are elementary divisor rings or not. In order to overcome this, we study different

properties that we can extend Bézout domains with to make them elementary divi-

sor rings. The properties we define and discuss in this section are:

1. Adequacy (i.e. the existence of a gdco operation);

2. Krull dimension ≤ 1;

3. Strict divisibility is well-founded (constructive principal ideal domains).

We have already considered the last one of these in section 3, but here we formalize

an alternative proof that constructive principal ideal domains are elementary divisor

rings, using a reduction due to Kaplansky [24]. It consists in first simplifying the

problem of computing Smith normal form for m× n matrices to the 2× 2 case and

then showing that any 2× 2 matrix has a Smith normal form if and only if the ring

satisfies the “Kaplansky condition”. This means that it suffices for us to prove that

the three different extensions all imply this condition in order to show that they are

elementary divisor rings.

5.1 The Kaplansky condition

The reduction of the computation of Smith normal form of arbitrary m×n matrices

to 2 × 2 matrices is done by extracting an algorithm from the proof of theorem

5.1 in [24]. The formalization is done by first implementing this algorithm, called

smithmxn, computing the Smith normal form of arbitrary sized matrices assuming

an operation computing it for 2× 2 matrices and then proving that this algorithm

satisfies smith_spec:

Lemma smithmxnP :

forall (smith2x2 : 'M[R]_2 -> 'M[R]_2 * seq R * 'M[R]_2),

(forall (M : 'M[R]_2), smith_spec M (smith2x2 M)) ->

forall m n (M : 'M[R]_(m,n)), smith_spec M (smithmxn smith2x2 M).

This algorithm has no assumptions on the underlying ring except that it is an inte-

gral domain. It can be generalized to arbitrary commutative rings but then we also

need to be able to put 1× 2 and 2× 1 matrices in Smith normal form.

Now consider a 2× 2 matrix
�

a b

c d

�

with coefficients in a Bézout domain. We can compute g = gcd(a, c) and a1 and c1

such that a = a1 g and c = c1 g. We also have u and v such that ua1+ vc1 = 1. Using

this we can form:

�

u v

−c1 a1

��

a b

c d

�

=

�

ua+ vc ub+ vd

−c1a+ a1c −c1 b+ a1d

�

=

�

ua+ vc ub+ vd

0 −c1 b+ a1d

�

So it suffices to consider matrices of the following shape:

�

a b

0 c

�

and without loss of generality we can assume that gcd(a, b, c) = 1. Now, such a

matrix has a Smith normal form if and only if it satisfies the Kaplansky condition:

for all a, b, c ∈ R with gcd(a, b, c) = 1 there exist p, q ∈ R with gcd(pa, pb+qc) = 1.

The interesting step for the reduction is the right to left direction of the “if and

only if”, so let us sketch how it is proved: assume that R is a Bézout domain that

satisfies the Kaplansky condition and consider an upper triangular matrix with ele-

ments a, b and c with gcd(a, b, c) = 1. From the Kaplansky condition we get p and

q such that gcd(pa, pb+qc) = 1. This means that we also have x1 and y1 such that

pax1 + (pb + qc)y1 = 1. By reorganizing this we get p(ax1 + b y1) + qc y1 = 1, let

x = ax1 + b y1 and y = c y1. We can form the product:

�

p q

−y x

��

a b

0 c

��

x1 pb+ qc

y1 −pa

�

=

�

1 0

0 −ac

�

In order to formalize this proof we assume that we have an operation taking a, b

and c computing p and q satisfying the Kaplansky condition:

Variable kap : R -> R -> R -> R * R.

Hypothesis kapP : forall (a b c : R), gcdr a (gcdr b c) %= 1 ->

let: (p,q) := kap a b c in coprimer (p * a) (p * b + q * c).

We then define a function kapW : R -> R -> R -> R * R to extract the two

witnesses x1 and y1 from above, i.e. x1 and y1 such that x1pa + y1(pb + qc) = 1.

To do this we first prove:

Lemma coprimerP (a b : R) :

reflect (exists (xy : R * R), xy.1 * a + xy.2 * b = 1) (coprimer a b).

and we can then define a function computing (x1, y1) by turning the existential

statement in coprimerP into a Σ-type (i.e. a dependent pair). More precisely, we

have defined it by:

Definition kapW a b c : R * R :=

let: (p,q) := kap a b c in

if coprimerP (p * a) (p * b + q * c) is ReflectT P

then projT1 (sig_eqW P) else (0,0).

Here sig_eqW is a function from the SSREFLECT library that transforms our exis-

tential statement into a Σ-type, the first component of the resulting Σ-type is then

extracted using projT1. This is possible because R is taken to be an SSREFLECT

“choice type”, i.e. a type with a choice operator.

Once we have defined kapW, we can easily write the function computing Smith

normal form of 2×2 matrices, called kap_smith, and prove that it satisfies smith_spec:

Definition kap_smith (M : 'M[R]_2) : 'M[R]_2 * seq R * 'M[R]_2 :=

let: A := Bezout_step (M 0 0) (M 1 0) M 0 in

let: a00 := A 0 0 in let: a01 := A 0 1 in let: a11 := A 1 1 in

let: (d,_,_,_,a,b,c) := egcdr3 a00 a01 a11 in

if d == 0 then (Bezout_mx (M 0 0) (M 1 0) 0,[::],1%:M) else

let: (p,q) := kap a b c in

let: (x1,y1) := kapW a b c in

let: (x,y) := (a * x1 + y1 * b, c * y1) in

(mx2 p q (- y) x *m Bezout_mx (M 0 0) (M 1 0) 0,

map (fun x => d * x) [:: 1; - a * c],

mx2 x1 (p * b + q * c) y1 (- p * a)).

Lemma kap_smithP (M : 'M[R]_2) : smith_spec M (kap_smith M).

Here mx2 is a notation to define 2 × 2 matrices and egcdr3 computes the Bézout

coefficients for 3 elements.

We have also formalized the other direction, so for a Bézout domain, satisfying

the Kaplansky condition is equivalent to being an elementary divisor ring. Hence it

suffices to prove that the various extensions to Bézout domains satisfy the Kaplansky

condition in order to get that they are elementary divisor rings.

5.2 The three extensions to Bézout domains

In this section we discuss three extensions to Bézout domains that imply the Ka-

plansky condition.

5.2.1 Adequate domains

In [21] Helmer introduced the notion of adequate domains. These are Bézout

domains where for any a, b ∈ R, with b 6= 0, there exists r ∈ R such that:

1. r | b,

2. r is coprime with a, and

3. for all non unit d such that dr | b we have that d is not coprime with a.

We have proved that this notion is equivalent to having a “gdco” function. This

function has previously been introduced by one of the authors in [6] in order to

implement quantifier elimination for algebraically closed fields. It has also other

applications in algebra, see [27]. It takes two elements a, b ∈ R, with b 6= 0, and

computes r such that:

1. r | b,

2. r is coprime with a , and

3. for all divisors d of b that is coprime to a we have d | r.

This means that r is the greatest divisor of b that is coprime to a. These notions are

expressed in COQ as:

Inductive adequate_spec (a b : R) : R -> Type :=

| AdequateSpec0 of b = 0 : adequate_spec a b 0

| AdequateSpec r of b != 0

& r %| b

& coprimer r a

& (forall d, d * r %| b -> d \isn't a GRing.unit ->

~~ coprimer d a)

: adequate_spec a b r.

Inductive gdco_spec (a b : R) : R -> Type :=

| GdcoSpec0 of b = 0 : gdco_spec a b 0

| GdcoSpec r of b != 0

& r %| b

& coprimer r a

& (forall d, d %| b -> coprimer d a -> d %| r)

: gdco_spec a b r.

Lemma adequate_gdco a b r : adequate_spec a b r -> gdco_spec a b r.

Lemma gdco_adequate a b r : gdco_spec a b r -> adequate_spec a b r.

We have implemented an algorithm called gdco_kap that computes p and q in

the Kaplansky condition using the gdco operation. Using this we have proved:

Lemma gdco_kapP (a b c : R) : gcdr a (gcdr b c) %= 1 ->

let: (p, q) := gdco_kap a b c in coprimer (p * a) (p * b + q * c).

Using this we can define a function that computes the Smith normal form for

any matrix over an adequate domain:

Definition gdco_smith := smithmxn (kap_smith gdco_kap).

Lemma gdco_smithP m n (M : 'M[R]_(m,n)) : smith_spec M (gdco_smith M).

Hence we get that adequate domains are elementary divisor rings.

5.2.2 Krull dimension ≤ 1

The next class of rings we study are Bézout domains of Krull dimension ≤ 1. Clas-

sically Krull dimension is defined as the supremum of the length of all chains of

prime ideals, this means that a ring has Krull dimension n ∈ N if there is a chain of

prime ideals:

p0 ( p1 ( · · · ( pn

but no such chain of length n+ 1. For example, a field has Krull dimension 0 and

any principal ideal domain (that is not a field) has Krull dimension 1. This can

be defined constructively using an inductive definition as in [25]. Concretely an

integral domain R is of Krull dimension ≤ 1 if for any a, u ∈ R there exists v ∈ R

and n ∈ N such that

a | un(1− uv)

In order to prove that Bézout domains of Krull dimension ≤ 1 are adequate we

first prove:

Hypothesis krull1 : forall a u, exists m v, a %| u ^+ m * (1 - u * v).

Lemma krull1_factor a b : exists n b1 b2,

[&& 0 < n, b == b1 * b2, coprimer b1 a & b2 %| a ^+ n].

This means that given a and b we can compute n ∈ N and b1, b2 ∈ R such that

n 6= 0, b = b1 b2, b1 is coprime with a and b2 | a
n. If we set r to b1 in the definition

of adequate domains we have to prove:

1. b1 | b1 b2,

2. b1 is coprime with a, and

3. for all non unit d such that d b1 | b1 b2 we have that d is not coprime with a.

The first two are obvious. For the third point, we have to prove that any non-unit

d that divides b2 is not coprime with a. So it suffices to prove that any d coprime

with a that divides b2 is a unit. Now as n 6= 0 we get that d is coprime with an, but

d | b2 and b2 | a
n so d must be a unit. We have formalized this argument in:

Lemma krull1_adequate a b : { r : R & adequate_spec a b r }.

This means that Bézout domains of Krull dimension≤ 1 are adequate and hence sat-

isfy the Kaplansky condition, which in turn means that they are elementary divisor

rings:

Definition krull1_gdco a b := projT1 (krull1_adequate a b).

Definition krull1_smith := gdco_smith krull1_gdco.

Lemma krull1_smithP m n (M : 'M[R]_(m,n)) : smith_spec M (krull1_smith M).

5.2.3 Constructive principal ideal domains

Finally, we have showed that constructive principal ideal domains are adequate

domains by proving that given a and b we can compute r satisfying gdco_spec:

Lemma pid_gdco (R : pidType) (a b : R) : {r : R & gdco_spec a b r}.

The construction of the greatest divisor of a coprime to b in a constructive prin-

cipal ideal domain is done as in the particular case of polynomials in [6]. If gcd(a, b)

is a unit, then a is trivially the result, otherwise we get a′ by dividing a by gcd(a, b)

and we repeat the process with a′ and b. This process terminates because when

gcd(a, b) is not a unit, a′ strictly divides a and by our definition of constructive

principal ideal domains, there cannot be an infinite decreasing sequence for strict

divisibility.

This way we get an alternative proof that constructive principal ideal domains

are elementary divisor rings:

Definition pid_smith := gdco_smith (fun a b => projT1 (pid_gdco a b)).

Lemma pid_smithP m n (M : 'M[R]_(m,n)) : smith_spec M (pid_smith M).

This proof is simpler in the sense that we first reduce the problem of computing the

Smith normal form to computing the gdco of two elements. This way, the part of

the proof based on well-founded recursion is concentrated to pid_gdco instead of

being interleaved in the algorithm computing the Smith normal form of arbitrary

m× n matrices.

6 Related work

Most proof systems have one or more libraries of formalized linear algebra. How-

ever, the specificity of our work is that it is more general than the usual study of

vector spaces (we do not require scalars to be in a field, but only in an elemen-

tary divisor ring) while still retaining an algorithmic basis, as opposed to a purely

abstract and axiomatized development. In particular, this work constitutes to our

knowledge the first formal verification of an algorithm for the Smith normal form

of matrices.

A fair amount of module theory and linear algebra has been formalized [34]

in MIZAR. But it is based on classical logic and does not account for underlying

algorithmic aspects. Likewise, a HOL LIGHT library [19] proves significant results

in linear algebra and on the topology of vector spaces, but it is specialized to Rn

and also classical.

Some other developments focus more on the algebra of vectors and matrices,

without providing support for point-free reasoning on subspaces. Let us cite [31] in

ISABELLE, which aims primarily to certify linear inequalities and [14, 22] in ACL2,

formalizing only matrix algebra.

In COQ too, older developments focus on the representation of matrices like [28],

or classical linear algebra over a field like [35], based on [33]. One exception is of

course the more recent work [16] we already mentioned and on which we based

this work, extending it from finitely generated vector spaces to finitely presented

modules over elementary divisor rings.

The authors are also developing a library of computational algebra called CO-

QEAL – the COQ Effective Algebra Library [5, 12]. It contains many examples of

algorithms from linear algebra like the rank of matrices over fields and Strassen’s

matrix multiplication [12], the Sasaki-Murao algorithm for computing the charac-

teristic polynomial of a matrix over a commutative ring [10], and the kernel of a

matrix over a field [23].

Two of the authors have previously formalized the theory of finitely presented

modules in COQ [7], building on a previous formalization of coherent and strongly

discrete rings [9] that provides a basis for a general treatment of matrix algebra.

The present work extends this to the theory of finitely presented modules over el-

ementary divisor rings, which gives a means for deciding whether two finitely pre-

sented modules are isomorphic or not as described in section 4.3. It also provides

concrete instances solving the basic algorithmic problems underlying the work on

finitely presented modules as elementary divisor rings provides interesting exam-

ples of coherent strongly discrete rings.

7 Conclusions and future work

The relationships between the notions introduced in this paper are depicted in fig-

ure 1. The numbers on the edges denote the sections in which the different impli-

cations and inclusions are proved:

Euclidean domain Coherent

PID EDR Bézout GCD domain

Kdim≤ 1 Adequate Strongly discrete

2
3.1

5.2.3

3.2

4.1

4.1

4.1

2

5.2.2

5.2.1
5

Figure 1: Relationship between the defined notions

The arrow between PID and Krull dimension ≤ 1 is dashed because it has not

been formally proved yet. A constructive proof of this can be found in [25]. We

currently see two options to formalize it: either we try to develop more extensively

the theory of ideals to stick close to the paper proof, or we expand statements on

ideal to statements on elements. Unlike the former, the latter option would require

no further infrastructure, but it is likely that the size of the proof would explode, as

in some proofs where we already had to talk about elements instead of ideals (e.g.

the lemma krull1_factor in the current state of the formalization).

It has been mentioned that Z and k[x] where k is a field are the basic examples

for all of these rings. Many more examples of Bézout domains are presented in

the chapters on Bézout domains and elementary divisor rings in [13] (for instance,

Bézout domains of arbitrary finite Krull dimension and an example of a Bézout

domain that is not adequate). It would be interesting see which of these could be

done in a constructive setting and formalize them in order to get more instances

than Z and k[x].

An important application of this work is to compute the homology of chain com-

plexes which provides a means to study properties of mathematical objects like

topological spaces. By computing homology one associates modules to these kinds

of objects, giving a way to distinguish between them. The Smith normal form of

matrices with coefficients in Z is at the heart of the computation of homology as

the universal coefficient theorem for homology [20] states that homology with co-

efficients in Z determines homology with coefficients in any other abelian group.

Note that the Kaplansky condition in section 5 is expressed using first-order

logic. It means that the open problem whether all Bézout domains are elementary

divisor rings can be expressed using first-order logic. We have formulated the prob-

lem this way and applied various automatic theorem provers in order to try to find

a proof that Bézout domains, alone, and with the two other assumptions (adequacy

or Krull dimension ≤ 1) are elementary divisor rings. However, none managed so

far.

We have in this paper presented the formalization of many results on elementary

divisor rings. This way we get interesting examples of coherent strongly discrete

rings and concrete algorithms for studying finitely presented modules. All of the

proofs have been performed in a constructive setting, and except for principal ideal

domains, without chain conditions.

